Laboratory 5: Higher order linear differential equations

The general form of an n -order linear differential equations is

$$y^{(n)}(x) + a_1(x)\, y^{(n-1)}(x) + \ \dots + a_n(x)\, y(x) = f(x)$$

Testing solutions

```
> with(DEtools):with(plots):
```

For a given differential equation you can check if some given function is or it is not a solution as follows:

```
> deq:=x*diff(y(x),x$2)-(x+3)*diff(y(x),x)+2*y(x)=0;
```

$$deq := x\left(\frac{d^2}{dx^2}\, y(x)\right) - (x+3)\left(\frac{d}{dx}\, y(x)\right) + 2\, y(x) = 0$$

Let's check if the function $\varphi(x) = x^2 + 4\,x + 6$ is a solution of the equation:

```
> varphi:=x->x^2+4*x+6;
```

$$\varphi := x \rightarrow x^2 + 4\,x + 6$$

```
> subs(y(x)=varphi(x),deq);
```

$$x\left(\frac{d^2}{dx^2}\left(x^2 + 4\,x + 6\right)\right) - (x+3)\left(\frac{d}{dx}\left(x^2 + 4\,x + 6\right)\right) + 2\,x^2 + 8\,x + 12 = 0$$

```
> simplify(%);
```

$$0 = 0$$

or we can use the **eval** command:

```
> eval(deq,y(x)=varphi(x));simplify(%);
```

$$10\,x - (x+3)\,(2\,x+4) + 2\,x^2 + 12 = 0$$

$$0 = 0$$

The **eval** and **subs** commands treat the evaluation of inert ODE differently: the **subs** command only inserts the given value(s) into the expression but does not perform any evaluation in the case of inert expressions, as the **eval** does.

So, indeed the function this function satisfies the differential equation.

For the function $\varphi(x) = e^{x^2}$ we get:

```
> phi:=x->exp(x^2);
```

$$\phi := x \rightarrow e^{x^2}$$

```
> subs(y(x)=phi(x),deq);
```

$$x\left(\frac{d^2}{dx^2}\, e^{x^2}\right) - (x+3)\left(\frac{d}{dx}\, e^{x^2}\right) + 2\, e^{x^2} = 0$$

```
> simplify(%);
```

$$2\,e^{x^2}\left(2x^3 - x^2 - 2x + 1\right) = 0$$

since we didn't get $0 = 0$ then this function does not satisfies the equation, so it is not a solution.

Also, we can use the **odetest** command to check if some function satisfies or not the given differential equation. The **odetest** command checks explicit and implicit solutions for ODEs by making a careful simplification of the ODE with respect to the given solution. If the solution is valid, the returned result will be 0; otherwise, the algebraic remaining expression will be returned.

```
> odetest(y(x)=varphi(x),deq,y(x));
```

$$0$$

```
> odetest(y(x)=phi(x),deq,y(x));
```

$$4\,e^{x^2}x^3 - 2x^2e^{x^2} - 4x\,e^{x^2} + 2\,e^{x^2}$$

Finding some particular solutions

Let's consider the ODE:

```
> deq:=x*diff(y(x),x$2)-(x+3)*diff(y(x),x)+2*y(x)=0;
```

$$deq := x\left(\frac{d^2}{dx^2}y(x)\right) - (x+3)\left(\frac{d}{dx}y(x)\right) + 2\,y(x) = 0$$

If we want to check if the ODE admits a solution of the form as a polynomial function of the 2nd degree, then:

```
> varphi:=x->a*x^2+b*x+c;
```

$$\varphi := x \to a\,x^2 + b\,x + c$$

```
> expr:=eval(deq,y(x)=varphi(x));
```

$$expr := 2\,x\,a - (x+3)\,(2\,a\,x + b) + 2\,a\,x^2 + 2\,b\,x + 2\,c = 0$$

```
> simplify(expr);
```

$$-4\,a\,x + b\,x - 3\,b + 2\,c = 0$$

The **collect** function views the left hand side of **expr** as a general polynomial in x. It collects all the coefficients with the same rational power of x. This includes positive and negative powers, and fractional powers.

```
> expr:=collect(lhs(expr),x);
```

$$expr := (-4\,a + b)\,x - 3\,b + 2\,c$$

The **coeff** function extracts the coefficient of x^n in the polynomial p.

```
> c1:=coeff(expr,x,1);
```

$$c1 := -4\,a + b$$

```
> c2:=coeff(expr,x,0);
```

$$c2 := -3\,b + 2\,c$$

```
> solve({c1=0,c2=0},{a,b,c});
```

$$\{a = a,\ b = 4\,a,\ c = 6\,a\}$$

So, indeed this equation admits as a solution a polynomial function of the form $\varphi(x) = a\,x^2 + 4\,x + 6\,a$, where a is a real parameter with $a \neq 0$. If we take a=1 we get $\varphi(x) = x^2 + 4\,x + 6$.

If we know that this equation admits a solution of the form $\varphi(x) = e^{a\,x}\,(b\,x + c)$, let's find this solution:

```
> varphi:=x->exp(a*x)*(b*x+c);
```

$$\varphi := x \to e^{a\,x}\,(b\,x + c)$$

```
> expr:=eval(deq,y(x)=varphi(x));
```

$$expr := x\left(a^2\,e^{x\,a}\,(b\,x + c) + 2\,a\,e^{x\,a}\,b\right) - (x + 3)\left(a\,e^{x\,a}\,(b\,x + c) + e^{x\,a}\,b\right) + 2\,e^{x\,a}\,(b\,x + c) = 0$$

```
> simplify(expr);
```

$$e^{x\,a}\left(a^2\,b\,x^2 + a^2\,c\,x - a\,b\,x^2 - a\,b\,x - a\,c\,x - 3\,a\,c + b\,x - 3\,b + 2\,c\right) = 0$$

The $e^{a\,x}$ cannot be 0, so the second factor must be 0. First, we need to simplify the left hand side of expr with $e^{a\,x}$

```
> expr:=exp(-a*x)*expr;
```

$$expr := e^{-x\,a}\left(x\left(a^2\,e^{x\,a}\,(b\,x + c) + 2\,a\,e^{x\,a}\,b\right) - (x + 3)\left(a\,e^{x\,a}\,(b\,x + c) + e^{x\,a}\,b\right) + 2\,e^{x\,a}\,(b\,x + c)\right) = 0$$

```
> expr:=simplify(expr);
```

$$expr := a^2\,b\,x^2 + a^2\,c\,x - a\,b\,x^2 - a\,b\,x - a\,c\,x - 3\,a\,c + b\,x - 3\,b + 2\,c = 0$$

```
> collect(expr,x);
```

$$\left(a^2\,b - a\,b\right)x^2 + \left(a^2\,c - a\,b - a\,c + b\right)x - 3\,a\,c - 3\,b + 2\,c = 0$$

```
> c1:=coeff(lhs(expr),x,2); c2:=coeff(lhs(expr),x,1);
c3:=coeff(lhs(expr),x,0);
```

$$c1 := a^2\,b - a\,b$$

$$c2 := a^2\,c - a\,b - a\,c + b$$

$$c3 := -3\,a\,c - 3\,b + 2\,c$$

```
> solve({c1=0,c2=0,c3=0},{a,b,c});
```

$$\{a = a,\ b = 0,\ c = 0\},\ \{a = 1,\ b = b,\ c = -3\,b\}$$

We get two possibilities: $\varphi(x) = 0$ or $\varphi(x) = e^x\,(b\,x - 3\,b)$. We know that the null function is always a solution of a linear homogeneous equation, but it cannot be used to construct a fundamental system of solution, so we take the second solution with $b = 1$, $\varphi(x) = e^x\,(x - 3)$:

```
> a:=1;b:=1;c:=-3*b;
```

$$a := 1$$

$$b := 1$$

$$c := -3$$

```
> varphi(x);
```

$$e^x\,(x - 3)$$

```
> odetest(y(x)=varphi(x),deq);
```

$$0$$

Fundamental system of solutions

Let S = $\{y_1(x), y_2(x), \ldots, y_n(x)\}$ be a set of n functions for which each is differentiable at least $n-1$ times. The Wronskian of system functions S, denoted by $W(x, y_1(x), y_2(x), \ldots, y_n(x))$ is the determinant of:

$$\begin{bmatrix} y_1(x) & y_2(x) & \ldots & y_n(x) \\ \frac{d}{dx} y_1(x) & \frac{d}{dx} y_2(x) & \ldots & \frac{d}{dx} y_n(x) \\ \ldots & \ldots & \ldots & \ldots \\ \frac{d^{n-1}}{dx^{n-1}} y_1(x) & \frac{d^{n-1}}{dx^{n-1}} y_2(x) & \ldots & \frac{d^{n-1}}{dx^{n-1}} y_n(x) \end{bmatrix}$$

The Wronskian criterion is used check if a given system of solutions $\{y_1(x), y_2(x), \ldots, y_n(x)\}$ is linearly independent or not. If the Wronskian is not 0 then the system of solution is a fundamental system of solutions, so the general solution of the linear homogeneous equation is a linear combination of these solutions.

Let's consider the following linear homogeneous ODE:

```
> deq:=x^2*diff(y(x),x$2)-2*x*diff(y(x),x)+2*y(x)=0;
```

$$deq := x^2 \left(\frac{d^2}{dx^2} y(x)\right) - 2\,x \left(\frac{d}{dx} y(x)\right) + 2\,y(x) = 0$$

Let's check if the functions system $\{x, x^2\}$ is a fundamental system of solutions. First, we check if these functions are solutions and second we check if the Wronskian is not 0.

```
> varphi[1]:=x->x;
```

$$\varphi_1 := x \rightarrow x$$

```
> varphi[2]:=x->x^2;
```

$$\varphi_2 := x \rightarrow x^2$$

```
> odetest(y(x)=varphi[1](x),deq);
```

$$0$$

```
> odetest(y(x)=varphi[2](x),deq);
```

$$0$$

or we can use the **eval** command:

```
> eval(deq,y(x)=varphi[1](x));
```

$$0 = 0$$

```
> eval(deq,y(x)=varphi[2](x));
```

$$0 = 0$$

Indeed, these functions are solutions. Next, we compute the Wronskian:

```
> with(linalg):
```

The **wronskia**n command belongs to the **linalg** package, so load it!

```
> A:=wronskian( [varphi[1](x),varphi[2](x)],x);
```

$$A := \begin{bmatrix} x & x^2 \\ 1 & 2\,x \end{bmatrix}$$

```
> det(A);
```

$$x^2$$

We get that the wronskian is x^2, so it is not 0 since we solve the equation on an interval which it does not contain the point $x = 0$, thus the given system is a fundamental system of solutions.

Constructing a homogeneous linear ODE for given solutions

It is possible to construct the homogeneous linear ODE when it is known the fundamental system of solutions. Suppose that the function system $\{y_1(x), y_2(x), ..., y_n(x)\}$ is a fundamental system of solutions for some homogeneous linear ODE. Taking any other solution y(x) of the equation then the system $\{y(x), y_1(x), y_2(x), ..., y_n(x)\}$ is linearly dependent, since $y(x)$ is a linear combination of the solutions from the fundamental system, so the Wronskian $W\big(x, y(x), y_1(x), y_2(x), ..., y_n(x)\big) = 0$ Using this property, we can obtain the ODE for which the system $\{y_1(x), y_2(x), ..., y_n(x)\}$ is a fundamental system of solutions.

For example, the functions system $\{x, x^2\}$ is linearly independent. Let's construct the corresponding linear homogeneous ODE for which this system is a fundamental system of solutions:

```
> varphi[1]:=x->x;
```

$$\varphi_1 := x \to x$$

```
> varphi[2]:=x->x^2;
```

$$\varphi_2 := x \to x^2$$

```
> A:=wronskian( [y(x),varphi[1](x),varphi[2](x)],x);
```

$$A := \begin{bmatrix} y(x) & x & x^2 \\ \frac{d}{dx}\,y(x) & 1 & 2\,x \\ \frac{d^2}{dx^2}\,y(x) & 0 & 2 \end{bmatrix}$$

```
> deq:=det(A)=0;
```

$$deq := x^2 \left(\frac{d^2}{dx^2}\,y(x)\right) - 2\,x\left(\frac{d}{dx}\,y(x)\right) + 2\,y(x) = 0$$

Solving a second order linear homogeneous ODE with nonconstant coefficients

We can solve a second order linear homogeneous ODE

$y''(x) + a_1(x)\, y'(x) + a_2(x)\, y(x) = 0$

if we know at least one solution φ(x). Using the substitution $y(x) = z(x)\, \varphi(x)$ we get a second order linear homogeneous equation of the form

$z''(x) + b_1(x)\, z'(x) = 0$

which admits the order reduction for $z'(x) = u(x)$ and we obtain a first order linear homogeneous ODE

$u'(x) + b_1(x)\, u(x) = 0$

Solving this equation, we get $u(x)$ and solving the equation $z'(x) = u(x)$ we get $z(x)$. Using the result, we obtain the solution $y(x)$.

Let's find the general solution of the ODE:

```
> deq:=x*diff(y(x),x$2)-(x+3)*diff(y(x),x)+2*y(x)=0;
```

$$deq := x\left(\frac{d^2}{dx^2}\, y(x)\right) - (x+3)\left(\frac{d}{dx}\, y(x)\right) + 2\, y(x) = 0$$

knowing that the function:

```
> varphi:=x->x^2+4*x+6;
```

$$\varphi := x \rightarrow x^2 + 4\, x + 6$$

is a solution.

```
> odetest(y(x)=varphi(x),deq);
```

$$0$$

```
> eval(deq,y(x)=z(x)*varphi(x));
```

$$x\left(\left(\frac{d^2}{dx^2}\, z(x)\right)\left(x^2 + 4\, x + 6\right) + 2\left(\frac{d}{dx}\, z(x)\right)(2\, x + 4) + 2\, z(x)\right) - (x+3)\left(\left(\frac{d}{dx}\, z(x)\right)\left(x^2 + 4\, x + 6\right) + z(x)\,(2\, x + 4)\right) + 2\, z(x)\left(x^2 + 4\, x + 6\right) = 0$$

```
> deq2:=simplify(%);
```

$$deq2 := \left(\frac{d^2}{dx^2} z(x)\right) x^3 - \left(\frac{d}{dx} z(x)\right) x^3 + 4 \left(\frac{d^2}{dx^2} z(x)\right) x^2 - 3 \left(\frac{d}{dx} z(x)\right) x^2 + 6 \left(\frac{d^2}{dx^2} z(x)\right) x - 10 \left(\frac{d}{dx} z(x)\right) x - 18 \left(\frac{d}{dx} z(x)\right) = 0$$

in order to see the coefficients of $\frac{d^2}{dx^2} z(x)$ and $\frac{d}{dx} z(x)$ we can use the **collect** command with option **{diff(z(x),x$2),diff(z(x),x)}**

```
> deq2:=collect(deq2,{diff(z(x),x$2),diff(z(x),x)});
```

$$deq2 := (x^3 + 4 x^2 + 6 x) \left(\frac{d^2}{dx^2} z(x)\right) + (-x^3 - 3 x^2 - 10 x - 18) \left(\frac{d}{dx} z(x)\right) = 0$$

```
> deq3:=subs(diff(z(x),x)=u(x),diff(z(x),x$2)=diff(u(x),x),deq2);
```

$$deq3 := (x^3 + 4 x^2 + 6 x) \left(\frac{d}{dx} u(x)\right) + (-x^3 - 3 x^2 - 10 x - 18) u(x) = 0$$

```
> sol1:=dsolve(deq3,u(x));
```

$$sol1 := u(x) = \frac{\_C1\, e^x x^3}{(x^2 + 4 x + 6)^2}$$

```
> uu:=unapply(rhs(sol1),x,_C1);
```

$$uu := (x, \_C1) \to \frac{\_C1\, e^x x^3}{(x^2 + 4 x + 6)^2}$$

```
> deq4:=diff(z(x),x)=uu(x,_C1);
```

$$deq4 := \frac{d}{dx} z(x) = \frac{\_C1\, e^x x^3}{(x^2 + 4 x + 6)^2}$$

```
> sol2:=dsolve(deq4,z(x));
```

$$sol2 := z(x) = \frac{(x - 3)\, \_C1\, e^x}{x^2 + 4 x + 6} + \_C2$$

```
> zz:=unapply(rhs(sol2),x,_C1,_C2);
```

$$zz := (x, \_C1, \_C2) \to \frac{(x - 3)\, \_C1\, e^x}{x^2 + 4 x + 6} + \_C2$$

```
> yy:=zz(x,_C1,_C2)*varphi(x);
```

$$yy := \left(\frac{(x - 3)\, \_C1\, e^x}{x^2 + 4 x + 6} + \_C2\right) (x^2 + 4 x + 6)$$

```
> yy:=simplify(yy);
```

$$yy := \_C1\, e^x x + \_C2\, x^2 - 3\, \_C1\, e^x + 4\, \_C2\, x + 6\, \_C2$$

```
> collect(yy,{_C1,_C2});
```

$$(e^x x - 3 e^x)\, \_C1 + (x^2 + 4 x + 6)\, \_C2$$

Variation of the constants method

The general solution of a linear nonhomogeneous differential equations

$y^{(n)}(x) + a_1(x)\, y^{(n-1)}(x) + \ \dots + a_n(x)\, y(x) = f(x)$

is

$y(x) = y_0(x) + y_p(x)$

where $y_0(x)$ is the general solution of the homogeneous equation

$y_p(x)$ is a particular solution of the nonhomogeneous equation

If we know a fundamental system of solution for the homogeneous equation, then we can find a particular solution $y_p(x)$ using the variation of the constants method. If $\{y_1(x), y_2(x),\dots, y_n(x)\}$ is a fundamental system of solution, we look after the particular solution of the form

$y_p(x) = \phi_1(x)\, y_1(x) + \dots + \phi_n(x)\, y_n(x)$

the unkown functions $\phi_1(x)$, ..., $\phi_n(x)$ can be determined, first we solve the system with the unkowns $\phi'_1(x)$, ..., $\phi'_n(x)$

$$\begin{bmatrix} y_1(x) & y_2(x) & \dots & y_4(x) \\ \frac{d}{dx} y_1(x) & \frac{d}{dx} y_2(x) & \dots & \frac{d}{dx} y_4(x) \\ \dots & \dots & \dots & \dots \\ \frac{d^{n-1}}{dx^{n-1}} y_1(x) & \frac{d^{n-1}}{dx^{n-1}} y_2(x) & \dots & \frac{d^3}{dx^3} y_4(x) \end{bmatrix} \begin{bmatrix} \phi'_1(x) \\ \phi'_2(x) \\ \dots \\ \phi_4(x) \end{bmatrix} = \begin{bmatrix} 0 \\ \dots \\ 0 \\ f(x) \end{bmatrix}$$

and then we integrate, and we get $\phi_1(x)$, ..., $\phi_n(x)$

Let's consider the ODE:

```
> deq:=diff(y(x),x$2)-2/x*diff(y(x),x)+2/x^2*y(x)=x*cos(x);
```

$$deq := \frac{d^2}{dx^2} y(x) - \frac{2\left(\frac{d}{dx} y(x)\right)}{x} + \frac{2\, y(x)}{x^2} = x\cos(x)$$

We know that he functions system $\{x, x^2\}$ is a fundamental system of solutions for the homogeneous equation.

```
> varphi[1]:=x->x;
```

$$\varphi_1 := x \to x$$

```
> varphi[2]:=x->x^2;
```

$$\varphi_2 := x \to x^2$$

The particular solution has the form

$y_p(x) = \phi_1(x)\, \varphi_1(x) + \phi_2(x)\, \varphi_2(x)$

We construct the Wronskian matrix

```
> A:=wronskian([varphi[1](x),varphi[2](x)],x);
```

$$A := \begin{bmatrix} x & x^2 \\ 1 & 2x \end{bmatrix}$$

and the column matrix

```
> B:=matrix([[diff(phi[1](x),x)],[diff(phi[2](x),x)]]);
```

$$B := \begin{bmatrix} \frac{d}{dx}\phi_1(x) \\ \frac{d}{dx}\phi_2(x) \end{bmatrix}$$

```
> f:=x->x*cos(x);
```

$$f := x \to x\cos(x)$$

```
> F:=matrix([[0],[f(x)]]);
```

$$F := \begin{bmatrix} 0 \\ x\cos(x) \end{bmatrix}$$

```
> lh:=evalm(A&*B);lh[1,1];lh[2,1];
```

$$lh := \begin{bmatrix} x\left(\frac{d}{dx}\phi_1(x)\right) + x^2\left(\frac{d}{dx}\phi_2(x)\right) \\ \frac{d}{dx}\phi_1(x) + 2x\left(\frac{d}{dx}\phi_2(x)\right) \end{bmatrix}$$

$$x\left(\frac{d}{dx}\phi_1(x)\right) + x^2\left(\frac{d}{dx}\phi_2(x)\right)$$

$$\frac{d}{dx}\phi_1(x) + 2x\left(\frac{d}{dx}\phi_2(x)\right)$$

```
> syst:=lh[1,1]=F[1,1],lh[2,1]=F[2,1];
```

$$syst := x\left(\frac{d}{dx}\phi_1(x)\right) + x^2\left(\frac{d}{dx}\phi_2(x)\right) = 0, \frac{d}{dx}\phi_1(x) + 2x\left(\frac{d}{dx}\phi_2(x)\right) = x\cos(x)$$

```
> syst2:=solve({syst},{diff(phi[1](x),x),diff(phi[2](x),x)});
```

$$syst2 := \left\{\frac{d}{dx}\phi_1(x) = -x\cos(x), \frac{d}{dx}\phi_2(x) = \cos(x)\right\}$$

```
> s1:=dsolve(syst2[1],phi[1](x));
```

$$s1 := \phi_1(x) = -\cos(x) - \sin(x)\,x + \_C1$$

```
> phi1:=unapply(rhs(s1),x,_C1);
```

$$\phi1 := (x, \_C1) \to -\cos(x) - \sin(x)\,x + \_C1$$

```
> s2:=dsolve(syst2[2],phi[2](x));
```

$$s2 := \phi_2(x) = \sin(x) + \_C1$$

```
> phi2:=unapply(rhs(s2),x,_C1);
```

$$\phi2 := (x, \_C1) \to \sin(x) + \_C1$$

We can take particular case of _C1=0, so we get as particular solution for the nonhomogeneous equation the function

```
> expr:=phi1(x,0)*varphi[1](x)+phi2(x,0)*varphi[2](x);
```

$$expr := (-\cos(x) - \sin(x)\,x)\,x + x^2 \sin(x)$$

```
> expr:=simplify(expr);
```

$$expr := -x\cos(x)$$

```
> yp:=unapply(expr,x);
```

$$yp := x \rightarrow -x\cos(x)$$

```
> odetest(y(x)=yp(x),deq);
```

$$0$$