Laboratory 2: Solving differential equations with MAPLE

Initialization

> **restart:**	clear previously stored values and variables from memory
> **with(DEtools):**	load the package for differential equations
> **with(plots):**	load the graphics package

The derivation operation (review)

The derivation of the functions can be made in two ways: using **diff** command or using the derivation operator **D**:

```
> f:=x->exp(x^2)+3;
```

$$f := x \to e^{x^2} + 3$$

The **diff** command executes the derivation of the given expression with respect to the specified variable. The derivation operator **D** returns the derivate as a function.

```
> diff(f(x),x);
```

$$2\,x\,e^{x^2}$$

the second order derivate is given by

```
> diff(f(x),x,x);
```

$$2\,e^{x^2} + 4\,x^2\,e^{x^2}$$

also we can use the option **t$n** to get n-order derivative

```
> diff(f(x),x$2);
```

$$2\,e^{x^2} + 4\,x^2\,e^{x^2}$$

```
> diff(f(x),x$3);
```

$$12\,x\,e^{x^2} + 8\,x^3\,e^{x^2}$$

Using the derivation operator:

```
> D(f)(x);
```

$$2\,x\,e^{x^2}$$

If we the value of the first derivative in **x=1** then we can use:

```
> D(f)(1);
```

$$2\,e$$

The second oder derivative of *f* using **D**:

```
> (D@D)(f)(x);
```

$$2\,e^{x^2} + 4\,x^2\,e^{x^2}$$

and the value of the second oder derivative in **x=1:**

```
> (D@D)(f)(1);
```

$$6\,e$$

For the case of higher order derivative we can use **(D@@n)** :

```
> (D@@2)(f)(x);
```

$$2\, e^{x^2} + 4\, x^2\, e^{x^2}$$

```
> (D@D@D)(f)(x);
```

$$12\, x\, e^{x^2} + 8\, x^3\, e^{x^2}$$

```
> (D@@3)(f)(x);
```

$$12\, x\, e^{x^2} + 8\, x^3\, e^{x^2}$$

Define and solve a first order DE

Let consider the differential equation $\frac{d}{dx}\, y(x) = k\, y(x)$ where k is a real coefficient. The differential equation can be defined in MAPLE as follows:

```
> diff_eq1:=diff(y(x),x) = k*y(x);
```

$$diff\_eq1 := \frac{d}{dx}\, y(x) = k\, y(x)$$

To obtain the general solution of the equation use command **dsolve(ODE, y(x), options)** :

```
> dsolve(diff_eq1,y(x));
```

$$y(x) = \_C1\, e^{k\, x}$$

The general solution is seen as an expression. Notice that the undetermined constant is called *_C1*
How can we manipulate this expression?
We can use the function definition command:

```
> sol:=(x,k,c)->c*exp(k*x);
```

$$sol := (x, k, c) \rightarrow c\, e^{k\, x}$$

If the expression of the solution is too complicated, we can use the command **rhs** (*right hand side*) and **unapply** in order to obtain the solution as a function

```
> re:=rhs(dsolve(diff_eq1,y(x)));
```

$$re := \_C1\, e^{k\, x}$$

Using the **unapply** command we transform the expression **re** into a function specifying the variables:

```
> sol1:=unapply(re,x,k,_C1);
```

$$sol1 := (x, k, \_C1) \rightarrow \_C1\, e^{k\, x}$$

and we get the same result.

The graphics of ODE solutions

Let suppose that k:=2. Then the corresponding general solution is:

```
> y:=(x,c)->sol(x,2,c);
```

$$y := (x, c) \rightarrow sol(x, 2, c)$$

To draw the solutions curves you just assign some values for the constant c. For example take c:=1 c:=2 and c:=-1

```
> plot([y(x,1),y(x,2),y(x,-1)],x=-2..2);
```



If you want to obtain the solutions with some specified colors use the command:

```
> plot([y(x,1),y(x,2),y(x,-1)],x=-2..2,color=[black,red,blue]);
```



Also, you can specify the window of the graphic:

```
> plot([y(x,1),y(x,2),y(x,-1)], x=-2..2, y=-10..10, color = [black, red, blue]);
```



Using this way of manipulation for the solution you can see also how the solution depends on the *k* parameter. Let us consider c:=1 and assign some values for the parameter *k*.

```
> y1:=(x,k)->sol(x,k,1);
```

$$y1 := (x, k) \rightarrow sol(x, k, 1)$$

```
> plot([y1(x,0.2),y(x,0.5),y(x,-1)], x=-2..2, y=-10..10, color=[black, red, blue]);
```



> y:='y'; we clear the stored values in **y** in order to use again this variable in a new differential equation.

$$y := y$$

In the case of the following differential equation $\left(3\,y(x)^2+e^x\right)\left(\frac{d}{dx}\,y(x)\right)+e^x\,(y(x)+1)+\cos(x)=0$ the solution is given in implicit form:

```
> ecdif2:=(3*y(x)^2+exp(x))*diff(y(x),x)+exp(x)*(y(x)+1)+cos(x) = 0;
```

$$ecdif2 := \left(3\,y(x)^2+e^x\right)\left(\frac{d}{dx}\,y(x)\right)+e^x\,(y(x)+1)+\cos(x)=0$$

```
> sol2:=dsolve(ecdif2,y(x),implicit);
```

$$sol2 := e^x\,y(x)+e^x+\sin(x)+y(x)^3+\_C1=0$$

In order to plot the solution graph, we have to use the **implicitplot** command. First, we construct the implicit equation solution, in this case the expression from the left-hand side of the equation, to do that we use the **lhs** (letf hand side) command:

```
> lh:=lhs(sol2);
```

$$lh := e^x\,y(x)+e^x+\sin(x)+y(x)^3+\_C1$$

Also, we have to substitute **y(x)** with **y** in this expression in order to get a proper function with respect to the variables **x** and **y**

```
> lh1:=subs( y(x)=y, lh );
```

$$lh1 := e^x\,y+e^x+\sin(x)+y^3+\_C1$$

```
> f:=unapply(lh1,x,y,_C1);
```

$$f := (x, y, \_C1) \rightarrow e^x\,y+e^x+\sin(x)+y^3+\_C1$$

```
> implicitplot([f(x,y,0)=0,f(x,y,0.5)=0,f(x,y,1)=0], x=-5..5, y=-5..5, numpoints=10000);
```



If we want to graph more solutions we can generate the corresponding functions sequence **`f(x,y,c)`**. For example, in the case of the values **c =-4, -19/5, ...,-1/5,   0, 1/5, 2/5, ..., 4** we construct the functions sequence using **`seq`** command:

```
> sol_seq:=seq(f(x,y,i/5)=0,i=-20..20);
```

$$sol\_seq := e^x y + e^x + \sin(x) + y^3 - 4 = 0,\ e^x y + e^x + \sin(x) + y^3 - \frac{19}{5} = 0,\ e^x y + e^x + \sin(x) + y^3 - \frac{18}{5} = 0,\ e^x y + e^x + \sin(x) + y^3 - \frac{17}{5} = 0,\ e^x y + e^x + \sin(x) + y^3 - \frac{16}{5} = 0,\ e^x y + e^x + \sin(x) + y^3 - 3 = 0,\ e^x y + e^x + \sin(x) + y^3 - \frac{14}{5} = 0,\ e^x y + e^x + \sin(x) + y^3 - \frac{13}{5} = 0,\ e^x y + e^x + \sin(x) + y^3 - \frac{12}{5} = 0,\ e^x y + e^x + \sin(x) + y^3 - \frac{11}{5} = 0,\ e^x y + e^x + \sin(x) + y^3 - 2 = 0,\ e^x y + e^x + \sin(x) + y^3 - \frac{9}{5} = 0,\ e^x y + e^x + \sin(x) + y^3 - \frac{8}{5} = 0,\ e^x y + e^x + \sin(x) + y^3 - \frac{7}{5} = 0,\ e^x y + e^x + \sin(x) + y^3 - \frac{6}{5} = 0,\ e^x y + e^x + \sin(x) + y^3 - 1 = 0,\ e^x y + e^x + \sin(x) + y^3 - \frac{4}{5} = 0,\ e^x y + e^x + \sin(x) + y^3 - \frac{3}{5} = 0,\ e^x y + e^x + \sin(x) + y^3 - \frac{2}{5} = 0,\ e^x y + e^x + \sin(x) + y^3 - \frac{1}{5} = 0,\ e^x y + e^x + \sin(x) + y^3 = 0,\ e^x y + e^x + \sin(x) + y^3 + \frac{1}{5} = 0,\ e^x y + e^x + \sin(x) + y^3 + \frac{2}{5} = 0,\ e^x y + e^x + \sin(x) + y^3 + \frac{3}{5} = 0,\ e^x y + e^x + \sin(x) + y^3 + \frac{4}{5} = 0,\ e^x y + e^x + \sin(x) + y^3 + 1 = 0,\ e^x y + e^x + \sin(x) + y^3 + \frac{6}{5} = 0,\ e^x y + e^x + \sin(x) + y^3 + \frac{7}{5} = 0,\ e^x y + e^x + \sin(x) + y^3 + \frac{8}{5} = 0,\ e^x y + e^x + \sin(x) + y^3 + \frac{9}{5} = 0,\ e^x y + e^x + \sin(x) + y^3 + 2 = 0,\ e^x y + e^x + \sin(x) + y^3 + \frac{11}{5} = 0,\ e^x y + e^x + \sin(x) + y^3 + \frac{12}{5} = 0,\ e^x y + e^x + \sin(x) + y^3 + \frac{13}{5} = 0,\ e^x y + e^x + \sin(x) + y^3 + \frac{14}{5} = 0,\ e^x y + e^x + \sin(x) + y^3 + 3 = 0,\ e^x y + e^x + \sin(x) + y^3 + \frac{16}{5} = 0,\ e^x y + e^x + \sin(x) + y^3 + \frac{17}{5} = 0,\ e^x y + e^x + \sin(x) + y^3 + \frac{18}{5} = 0,\ e^x y + e^x + \sin(x) + y^3 + \frac{19}{5} = 0,\ e^x y + e^x + \sin(x) + y^3 + 4 = 0$$

```
> implicitplot([sol_seq],x=-5..5,y=-5..5,numpoints=10000);
```



Solving an IVP for a first order DE

Suppose that we want to solve the IVP $\frac{d}{dx} y(x) = k\, y(x)$ with the initial condition **y(0)=1**. The structure of the dsolve command is **dsolve({ODE, ICs}, y(x), options)**

```
> restart:with(DEtools):
> diff_eq:=diff(y(x),x) = k*y(x);
```

$$diff\_eq := \frac{d}{dx} y(x) = k\, y(x)$$

```
> in_cond:=y(0)=1;
```

$$in\_cond := y(0) = 1$$

```
> dsolve({diff_eq,in_cond},y(x));
```

$$y(x) = e^{k\, x}$$

Let us consider the case k:=2

```
> k:=2;
```

$$k := 2$$

```
> sol:=dsolve({diff_eq,in_cond},y(x));
```

$$sol := y(x) = e^{2\, x}$$

```
> yy:=x->(rhs(sol));
```

$$yy := x \rightarrow rhs(sol)$$

```
> plot(yy(x),x=-1..1,y=0..4);
```



You can obtain the graph the IVP directly using the command **`DEplot`**, the structure is **`DEplot(deqns, vars, trange, inits, options)`**:

```
> DEplot(diff_eq,y(x),x=-1..1,[[in_cond]]);
```



In this graph is also represented the direction field of the equation. If you want the graphs of the solutions for different initial condition (**`y(0)=1`**, **`y(0)=1.5`**, **`y(0)=2`**) you can use the same command and specify the list of the initial conditions:

```
> DEplot(diff_eq,y(x),x=-1..1,[[y(0)=1],[y(0)=1.5],[y(0)=2]]);
```



Solving a second order DE

```
> restart:
> with(DEtools):
> with(plots):
```

Warning, the name changecoords has been redefined

Consider the linear differential equation with the constant coeficients **y''+3y'+2y=1+x^2**

```
> deq1:=diff(y(x),x$2)+3*diff(y(x),x)+2*y(x)=1+x^2;
```

$$deq1 := \frac{d^2}{dx^2}\,y(x) + 3\left(\frac{d}{dx}\,y(x)\right) + 2\,y(x) = x^2 + 1$$

To obtain the general solution we use the **dsolve** command

```
> dsolve(deq1,y(x));
```

$$y(x) = \frac{1}{2}\,x^2 - \frac{3}{2}\,x + \frac{9}{4} - e^{-2\,x}\,\_C1 + e^{-x}\,\_C2$$

```
>
```

If we want to study the solution, we can use the same technique as in the previous section in order to draw the solution graph.

```
> sol:=dsolve(deq1,y(x));
```

$$sol := y(x) = \frac{1}{2}\,x^2 - \frac{3}{2}\,x + \frac{9}{4} - e^{-2\,x}\,\_C1 + e^{-x}\,\_C2$$

```
> right_hand:=rhs(sol);
```

$$right\_hand := \frac{1}{2}\,x^2 - \frac{3}{2}\,x + \frac{9}{4} - e^{-2\,x}\,\_C1 + e^{-x}\,\_C2$$

```
> y_sol:=unapply(right_hand,x,_C1,_C2);
```

$$y\_sol := (x, \_C1, \_C2) \rightarrow \frac{1}{2}\,x^2 - \frac{3}{2}\,x + \frac{9}{4} - e^{-2\,x}\,\_C1 + e^{-x}\,\_C2$$

Now we are able to one ore more than one solution graphs using the **plot** command.

```
> plot([y_sol(x,0,0),y_sol(x,0,1),y_sol(x,1,0)],x=-2..2,y=-10..10);
```



Solving an IVP for a second order DE

In the case of initial value problem for a second order differential equation we have two initial conditions, for example lets take **y(0)=1** and **y'(0)=0**.

REMARK: To define the condition **y'(0)=0** you must use the operator `D` !!!!

Let us consider the same ODE as in the previous section

```
> deq1:=diff(y(x),x$2)+3*diff(y(x),x)+2*y(x)=1+x^2;
```

$$deq1 := \frac{d^2}{dx^2}\,y(x) + 3\left(\frac{d}{dx}\,y(x)\right) + 2\,y(x) = x^2 + 1$$

```
> in_cond:=y(0)=1,D(y)(0)=0;
```

$$in\_cond := y(0) = 1,\ D(y)(0) = 0$$

To obtain the corresponding solution we use **dsolve** command in the following form:

```
> dsolve({deq1,in_cond},y(x));
```

$$y(x) = \frac{1}{2}\,x^2 - \frac{3}{2}\,x + \frac{9}{4} - \frac{1}{4}\,e^{-2\,x} - e^{-x}$$

Now we can use the previous technique (**rhs** and **unapply** comands) to construct the solution as a function and after that to represent its graph or we can obtain this graph directly using **DEplot** command.

```
> DEplot(deq1,y(x),x=-2..2,y=-10..10,[[in_cond]]);
```



If we need to draw more than one solution corresponding to different initial value problem we can use the same DEplot command specifying the list of initial conditions:

```
> DEplot(deq1,y(x),x=-2..2, [[y(0)=1,D(y)(0)=0], [y(0)=1,D(y)(0)=1],
[y(0)=0,D(y)(0)=0]] ,y=-10..10, color=red);
```



REMARK: Maple is unable to solve most of the second-order DE's explicitly. For information on numerically solving DE's, see Numerical Solutions with dsolve.
Consider the differential equation y''+xy'(x)+y(x)=sin(x). Try to use the dsolve command.

```
> deq2:=diff(y(x),x$2)+x*diff(y(x),x)+y(x)=sin(x);
```

$$deq2 := \frac{d^2}{dx^2} y(x) + x \left(\frac{d}{dx} y(x) \right) + y(x) = \sin(x)$$

```
> dsolve(deq2,y(x));
```

$$y(x) = e^{-\frac{1}{2}x^2} \operatorname{erf}\left(\frac{1}{2} I \sqrt{2}\, x\right) \_C1 + e^{-\frac{1}{2}x^2} \_C2$$
$$+ \frac{1}{4} I \sqrt{\pi}\, e^{\frac{1}{2}} \sqrt{2} \left(\operatorname{erf}\left(\frac{1}{2} I \sqrt{2}\, x - \frac{1}{2}\sqrt{2}\right)\right.$$
$$\left. + \operatorname{erf}\left(\frac{1}{2} I \sqrt{2}\, x + \frac{1}{2}\sqrt{2}\right)\right) e^{-\frac{1}{2}x^2}$$

```
> in_cond2:=y(0)=1,D(y)(0)=1;
```

$$in\_cond2 := y(0) = 1, D(y)(0) = 1$$

```
> dsolve({deq2,in_cond2},y(x));
```

$$y(x) = -\frac{1}{2} I e^{-\frac{1}{2}x^2} \operatorname{erf}\left(\frac{1}{2} I \sqrt{2}\, x\right) \sqrt{\pi} \left(e^{\frac{1}{2}} e^{-\frac{1}{2}} + 1\right) \sqrt{2}$$
$$+ e^{-\frac{1}{2}x^2} + \frac{1}{4} I \sqrt{\pi}\, e^{\frac{1}{2}} \sqrt{2} \left(\operatorname{erf}\left(\frac{1}{2} I \sqrt{2}\, x - \frac{1}{2}\sqrt{2}\right)\right.$$
$$\left. + \operatorname{erf}\left(\frac{1}{2} I \sqrt{2}\, x + \frac{1}{2}\sqrt{2}\right)\right) e^{-\frac{1}{2}x^2}$$

Maple expresses the solution in terms of the **erf** function.

$$\operatorname{erf}(x) = \frac{2\left(\int_0^x e^{-t^2}\, dt\right)}{\sqrt{\pi}}$$

We can obtain the numerical solution using in the dsolve command the option **type=numeric** and the **odeplot** comand to draw the corresponding graph.

```
> num_sol:=dsolve({deq2,in_cond2},y(x),type=numeric):
> odeplot(num_sol);
```