System of differential equations

In this chapter you will learn some tools for studying first order systems of differential equations using Maple 8. We will learn how to plot direction fields for first order systems of equations. Solution curves can also be added to these plots.

Solving a system of differential equation

The next example shows how Maple allows us to study a system of differential equations. We study the system:

x' = x+y
y' = x-y

```
> restart;
> eq1:=diff(x(t),t)=x(t)+y(t);
```

$$eq1 := \frac{d}{dt}x(t) = x(t) + y(t)$$

```
> eq2:=diff(y(t),t)=x(t)-y(t);
```

$$eq2 := \frac{d}{dt}y(t) = x(t) - y(t)$$

Again in this example we use the **DEtools** procedure **DEplot**.

```
> with(DEtools): with(plots):
> syst:=eq1,eq2;
```

$$syst := \frac{d}{dt}x(t) = x(t) + y(t),\ \frac{d}{dt}y(t) = x(t) - y(t)$$

```
> dsolve({syst},{x(t),y(t)});
```

$$\{x(t) = \_C1\, e^{\sqrt{2}\, t} + \_C2\, e^{-\sqrt{2}\, t}, y(t) = \_C1 \sqrt{2}\, e^{\sqrt{2}\, t} - \_C2 \sqrt{2}\, e^{-\sqrt{2}\, t} - \_C1\, e^{\sqrt{2}\, t} - \_C2\, e^{-\sqrt{2}\, t}\}$$

For the graphical representation of the solutions, we can construct the solutions functions using the **rhs** and **unapply** commands and then the values of the two integration constants _C1 and _C2 are given or we can use the **DEplot** command

```
> sol:=dsolve({syst},{x(t),y(t)});
```

$$sol := \{x(t) = \_C1\, e^{\sqrt{2}\, t} + \_C2\, e^{-\sqrt{2}\, t}, y(t) = \_C1 \sqrt{2}\, e^{\sqrt{2}\, t} - \_C2 \sqrt{2}\, e^{-\sqrt{2}\, t} - \_C1\, e^{\sqrt{2}\, t} - \_C2\, e^{-\sqrt{2}\, t}\}$$

The *sol* variable is a list, the access to x(t) and y(t) value can be made using **sol[1]** and **sol[2]**

```
> sol[1];
```

$$x(t) = \_C1\, e^{\sqrt{2}\, t} + \_C2\, e^{-\sqrt{2}\, t}$$

```
> sol[2];
```

$$y(t) = \_C1 \sqrt{2}\, e^{\sqrt{2}\, t} - \_C2 \sqrt{2}\, e^{-\sqrt{2}\, t} - \_C1\, e^{\sqrt{2}\, t} - \_C2\, e^{-\sqrt{2}\, t}$$

```
> xx:=unapply(rhs(sol[1]),t,_C1,_C2);
```

$$xx := (t, \_C1, \_C2) \to \_C1\, e^{\sqrt{2}\, t} + \_C2\, e^{-\sqrt{2}\, t}$$

```
> yy:=unapply(rhs(sol[2]),t,_C1,_C2);
```

$$yy := (t, \_C1, \_C2) \to \_C1\sqrt{2}\, e^{\sqrt{2}\, t} - \_C2\sqrt{2}\, e^{-\sqrt{2}\, t} - \_C1\, e^{\sqrt{2}\, t} - \_C2\, e^{-\sqrt{2}\, t}$$

In the variables **xx** and **yy** we have the solutions expressions as functions that depend on the independent variable t and the two integration constants $\_C1, \_C2$. If we want to obtain the graphs of the solutions for the constants **_C1 = 1** and **_C2 = 1** we use the **plot** command:

```
> plot(xx(t,1,1),t=-4..4);
```




```
> plot(yy(t,1,1),t=-4..4);
```



or both solutions can be represented in the same graph:

```
> plot([xx(t,1,1),yy(t,1,1)],t=-4..4,color=[red,blue]);
```



Initial value problem (Cauchy problem)

If we want to obtain the solutions corresponding to a initial value problem, for example:
x' = x+y
y' = x-y
x(0) = 1
y(0) = 0
the initial conditions are defined as in the case of differential equations:

```
> in_cond:=x(0)=1, y(0)=0;
```

$$in\_cond := x(0) = 1, y(0) = 0$$

```
> sol:=dsolve({syst,in_cond},{x(t),y(t)});
```

$$sol := \left\{x(t) = \left(\frac{1}{2} + \frac{1}{4}\sqrt{2}\right) e^{\sqrt{2}\,t} + \left(\frac{1}{2} - \frac{1}{4}\sqrt{2}\right) e^{-\sqrt{2}\,t}, y(t) = \left(\frac{1}{2} + \frac{1}{4}\sqrt{2}\right)\sqrt{2}\, e^{\sqrt{2}\,t} - \left(\frac{1}{2} - \frac{1}{4}\sqrt{2}\right)\sqrt{2}\, e^{-\sqrt{2}\,t} - \left(\frac{1}{2} + \frac{1}{4}\sqrt{2}\right) e^{\sqrt{2}\,t} - \left(\frac{1}{2} - \frac{1}{4}\sqrt{2}\right) e^{-\sqrt{2}\,t}\right\}$$

For the graphical representation of the solutions we have two alternatives. The first option is to define the solution functions using the `rhs` and `unapply` commands (as in the case of the general solution):

```
> xx:=unapply(rhs(sol[1]),t);
```

$$xx := t \to \left(\frac{1}{2} + \frac{1}{4}\sqrt{2}\right) e^{\sqrt{2}\,t} + \left(\frac{1}{2} - \frac{1}{4}\sqrt{2}\right) e^{-\sqrt{2}\,t}$$

```
> yy:=unapply(rhs(sol[2]),t);
```

$$yy := t \to \left(\frac{1}{2} + \frac{1}{4}\sqrt{2}\right)\sqrt{2}\, e^{\sqrt{2}\,t} - \left(\frac{1}{2} - \frac{1}{4}\sqrt{2}\right)\sqrt{2}\, e^{-\sqrt{2}\,t} - \left(\frac{1}{2} + \frac{1}{4}\sqrt{2}\right) e^{\sqrt{2}\,t} - \left(\frac{1}{2} - \frac{1}{4}\sqrt{2}\right) e^{-\sqrt{2}\,t}$$

and then we use the `plot` command

```
> plot(xx(t),t=-5..5);
```




```
> plot(yy(t),t=-5..5);
```



or both solutions can be represented in the same graph:

```
> plot([xx(t),yy(t)],t=-5..5,color=[red,blue]);
```



The second option of graphical representation is to use the **DEplot** command with the **scenes** option:

```
> DEplot([syst],[x,y],t=-5..5,[[in_cond]],linecolor=blue,scene=[t,x(t)]);
```




```
> DEplot([syst],[x,y],t=-5..5,[[in_cond]],linecolor=red,scene=[t,y(t)]);
```



If we want the graphical representation of both solutions in the same graph using the DEplot command we will use the instructions:

```
> xx1:=DEplot([syst],[x,y],t=-5..5,[[in_cond]],linecolor=blue,scene=[t,x(t)]):
> yy1:=DEplot([syst],[x,y],t=-5..5,[[in_cond]],linecolor=red,scene=[t,y(t)]):
> display([xx1,yy1]);
```



Direction field. Phase portrait

For a planar system (systems with two ODEs), the phase portrait is a representative set the solutions, plotted as parametric curves (with t as the parameter) on the Cartesian plane tracing the path of each particular solution (x, y) = (x(t), y(t)), $-\infty < t < \infty$. In this context, the Cartesian plane where the phase portrait resides is called the phase plane. The parametric curves traced by the solutions are sometimes also called their trajectories or orbits. A phase portrait is a graphical tool to visualize how the solutions of a given system of differential equations would behave in the long run.

The direction field or slope field is the set of slopes at the systems trajectories, a phase portrait is a graphical tool to visualize how the solutions of a given system of differential equations would behave in the long run.

To represent the direction field we use the instruction DEplot:

```
> DEplot([syst],[x(t),y(t)],t=-4..4,x=-10..10,y=-10..10,
arrows=medium);
```



To draw the phase portrait we use tha same command but we take in addition some initial conditions to draw the corresponding orbits

```
>
DEplot([syst],[x(t),y(t)],t=-4..4,x=-10..10,y=-10..10,[[in_cond]],
arrows=medium, linecolor=blue);
```



In this graph we have only one orbits. If we want more just give more initial conditions

```
> in_cond2:=[x(0)=0,y(0)=i]$i=1..5;
```

$$in\_cond2 := [x(0)=0, y(0)=1], [x(0)=0, y(0)=2], [x(0)=0, y(0)=3], [x(0)=0, y(0)=4], [x(0)=0, y(0)=5]$$

```
> DEplot([syst],[x(t),y(t)],t=-5..5,x=-10..10,y=-10..10,[in_cond2],
arrows=medium, linecolor=blue);
```



```
> in_cond3:=[x(0)=0,y(0)=i]$i=1..5,[x(0)=0,y(0)=(-1)*i]$i=1..5;
```

$$in\_cond3 := [x(0)=0, y(0)=1], [x(0)=0, y(0)=2], [x(0)=0, y(0)=3], [x(0)=0, y(0)=4], [x(0)=0, y(0)=5], [x(0)=0, y(0)=-1], [x(0)=0, y(0)=-2], [x(0)=0, y(0)=-3], [x(0)=0, y(0)=-4], [x(0)=0, y(0)=-5]$$

```
> DEplot([syst],[x(t),y(t)],t=-5..5,x=-10..10,y=-10..10,[in_cond3],
arrows=medium, linecolor=blue);
```